Spector Chapter 6.

Loops - Executing collections of statements repetitively.
The “DO Index=” statement

This is a standard looping statement. It’s best defined by example

Examples


data indexdo;

DO I=1 TO 10;

PUT I;

END;

PUT “Value of I after loop” I;

RUN;

data doexamp;

put "I've got a million things to do";

do i=1 to 1000000;

*y = sqrt(i);
if i > 10000 then if mod(i,10000)=0 then put i@;

end;

put;

put "I'm done.";

run;
Note:  PUT prints what follows to the current output file, the log file if you haven’t specified anything else.
Adding FILE PRINT; to the program causes PUT to send its stuff to the output window.

Check how quickly your computer executed this program, with and without the 
Mine executes the program in .12 seconds with the Y= statement and in .09 seconds with it comment’d.

A table of square roots
data roots;

do i=1 to 300;

root2=sqrt(i);

root3=i**(1/3);

put @1 i @10 root2 @25 root3;

end;

run;

A Coin flip program.

data coinflips;

nheads=0; ntails=0;

do i=1 to 100;

x=ranuni(-1);

if x <= .5 then nheads=nheads+1;

else ntails=ntails+1;

end;

put nheads " Heads and " ntails " Tails";

run;
Your own table of critical-values

There are some Inverse probability functions built into SAS.

They return the value of the statistic whose p-value is specified.

For example what’s the Z whose p-value is .05? 

What’s the chi-square value with df=10 which is just significant at .05?

Mike- First do this with an index from 1 to 20.

then justify the following program by the need for more df values, but

at larger intervals.

data critchis;

put "df" @10 ".05" @20 ".01" @30 ".001";

df = 0;

do df=1 to 100;

crit05 = round(cinv(.95,df),.001);

crit01 = round(cinv(.99,df),.001);

crit001 = round(cinv(.999,df),.001);

put df @10 crit05 @20 crit01 @30 crit001;

end;

run;
DO WHILE (Condition);

Within parentheses, a condition is defined that can be true or false.

The loop is executed as long as that condition holds.  It is checked at the END; statement.
It’s up to the programmer to include statements within the loop that will eventually result in the condition being false.  Otherwise, program will run forever.

data dowhile;

i=0;

do while (i<= 10);

i = i + 1;

put i;

end;   *This statement checks to see if condition is true.
put i;

run;

data critchis;

put "df" @10 ".05" @20 ".01" @30 ".001";

df = 0;

do while (df < 100);

if df < 20 then df=df+1;

else df=df+10;

crit05 = round(cinv(.95,df),.001);

crit01 = round(cinv(.99,df),.001);

crit001 = round(cinv(.999,df),.001);

put df @10 crit05 @20 crit01 @30 crit001;

end;

run;

DO UNTIL (Condition)
This form executes until a condition is met at the END; statement.

Not different in principle from WHILE, but sometimes it’s easier to turn right than it is to turn left.  SAS allows you to do both.

Nested loops

Loops within loops.

A multiplication table program.

data multtabl;

do i=1 to 10;

put;

do j=1 to 10;

prod = i*j;

col=5*(j-1);

put @col prod @;

end;

end;

run;

A multiple sample program.

data multsamp;

do sample = 1 to 100;

sum = 0;

do case = 1 to 20;

Xij = rannor(-1);

sum = sum + Xij;

end;

mean = round(sum/20,.01);

mean = 100*mean + 50;

put mean " " @;

output;    *Writes values to output file;

end;

proc freq data=multsamp;

 tables mean;

run;
Generates correlated scores.
data monte;

* Uses a well-known (among statisticians) result:

* If X is N(0,1), then 

*     Y = r*X + sqrt(1-r**2)*N(0,1)

* is N(0,1) approximately and 

* correlated approximately r with X;
do cases=1 to 100;

x = rannor(-1);

y = .7*x + sqrt(1-.7**2)*rannor(-1);

output;

end;

cards;

proc corr data=monte;
 var x; with y;

run;

data tsim;  * Spector’s Program 6.3, p 123;
********************** PARAMETERS**********************

it = Number of iterations of the simulation

plevel = Significnace level for the t-tests

meana = Population mean for group A

meanb = Population mean for group B

sda = Population standard deviation for group A

sdb = Population standard deviation for group B

na = Sample size for group A

nb = sample size for group B

********************************************************;
*Set parameter values;
it = 1000;

plevel = .05;

meana = 100; meanb = 100;

sda = 10; sdb = 100;

na = 20; nb = 5;

* initialize accumulators for mean of t's and number of significant t's;
tott = 0; nsig = 0;

*Loop to determine number of iterations;
do i=1 to it;

*Initialize accumulators;
sumxa = 0; sumxb= 0; sumxa2 = 0; sumxb2 = 0;

*Generate and accumulate group A observations;
do j=1 to na;

xa = meana + sda*rannor(0);

sumxa = sumxa + xa;

sumxa2 = sumxa2 + xa**2;

end;

*Generate and accumulate group B osservations;
do j=1 to nb;

xb = meanb + sdb*rannor(0);

sumxb = sumxb + xb;

sumxb2 = sumxb2 + xb**2;

end;

*Compute t and its significance;
ssa = sumxa2 - (sumxa**2 / na);

ssb = sumxb2 - (sumxb**2 / nb);

t = (sumxa/na - sumxb/nb)/ sqrt((ssa+ssb)/(na+nb-2)*(1/na+1/nb));

df = na + nb - 2;

pt = 2 * probt((-1*abs(t)),df);

*accumulate t's and significances;
tott = tott + it;

if pt < plevel then nsig = nsig + 1;

end;

*Calculate mean of t's and percent that were significant;
meant = tott/it;

percig = 100 * nsig / it;

*produce report of results;
*file print;
put @20 'Results of T-test Simulatio. ' / /;

put @10 nsig 'of ' it '(' percig '%) t-tests were significant at p < ' plevel;

put @10 'Means for the two populations = ' meana 'and ' meanb;

put @10 'Standard deviations for the two groups = ' sda 'and ' sdb;

put @10 'Samples sizes for the two groups = ' na 'and ' nb;

run;

Spector’s Program 7.1.  

*Program to score a psychological test.;
data score;

input id 2. (item1-item20) (1.);

array x(20) item1-item20;

do i=1 to 20;

put x(i)@;

end;

put ' ';

*reverse score appropriate items;
do i=1, 3, 5, 7, 9;

x(i) = 6 - x(i);

end;

do i= 11, 12, 13, 15;

x(i) = 8 - x(i);

end;

*convert missing values to 0;
do i=1 to 20;

if x(i) = . then x(i) = 0;

end;

*initialize item counters;
k1 = 0; k2=0; k3=0; *N's of nonmissing;
*initialize score accumulators;
scale1=0; scale2=0; scale3=0;

*sum items in each subscale;
do i=1 to 10;

scale1 = scale1 + x(i);

if x(i) > 0 then k1 = k1 + 1;

end;

do i= 11, 13, 19, 20;

scale2 = scale2 + x(i);

if x(i) > 0 then k2 = k2 + 1;

end;

do i=12, 14, 15, 16, 17, 18;

scale3 = scale3 + x(i);

if x(i) > 0 then k3 = k3 + 1;

end;

*Adjust scale totals for missing items;
* adj'd score = mean of nonmissing*no. nonmissing * scale length;
scale1 = scale1/k1*10;

scale2 = scale2/k2*4;

scale3 = scale3/k3*6;

*Round scores since might have dec cause of means;
scale1=round(scale1,.1);

scale2=round(scale2,.1);

scale3=round(scale3,.1);

*Eleminate scores based on half items or fewer;
if k1<6 then scale1=.;

if k2<3 then scale2=.;

if k3<4 then scale3=.;

cards;

0115342415241243276775

0251514233527767511243

0333332424247665632322

041 3 1 254576 532112

0515    2 33765

0613234353335554565565

0715253424331212566765

08 152515243534 7765

;

proc print;

run;

The output does not agree with Spector’s on p. 137.  Can’t find the mistake.
The series of statements is executed beginning with I=1 and ending with I=10;


I will equal 11 after the END statement.





Output





Note that I is set to 11 within the loop;





The check is at the END; statement.

